#version 3.5;
#include "param.inc"

global_settings {assumed_gamma 1.0}
camera {location <0,0,-5> look_at <1.3,0.9, 0>} 
light_source { <500,500,-500> rgb <1,0.9,0.8> }
light_source { <-100,100,-500> rgb <0.3,0.3,0.5> }

//dini
#declare a= 1;
#declare b= 0.2;
#declare F1= function(u,v){a*cos(u)*sin(v)}
#declare F2= function(u,v){a*sin(u)*sin(v)}
#declare F3= function(u,v){a*(cos(v)+ln(tan((v/2))))+b*u}

object{ 
   Parametric(
      F1, F2, F3,
      <0, FromV(0)>,
      <6*pi, 2>,
      120, 50,""
   )
   uv_mapping
   pigment {
      marble
      turbulence <0.15,0,0>
      translate <-0.5,0,0>
      scale <0.125,1,1>
      rotate <0,0,90>
      warp {repeat 0.5*x flip x}
      warp {repeat 0.5*y flip y}
      colour_map {
         [0, rgbf 0]
         [1, rgbf <1,1,1,0.6>]
      }
   }
   finish {
      specular 0.4
      roughness 0.05
      irid {
         0.3
         thickness 0.2
         turbulence 0.1
      }      
   }
   rotate <-90,45,0>
   rotate <-50,0,0>
   rotate <0,0,-50>
}